
1. Uvod

GPSS/FON predstavlja odličnu implementaciju GPSS simulacionog jezika. U prvoj verziji, a i ostalim doradama je bilo nekih nedostataka, a u ovom radu je pokušano da se bar neki od tih nedostataka otklone i samim tim omogući bolji i komforniji rad u verziji 4.0 ovog jezika. Ova verzija je nastala nadogradnjom verzije 3.04 od koje je mnogo naprednija. Rad je urađen u Borland Delphi 5 programskom okruženju u kome je urađena i prethodna implementacija. Sve novine koje su unete u GPSS/FON v 4.0 biće objašnjene u daljem tekstu
2. Promena implementacije GPSS/FON asemblera i procesora
U ranijim verzijama programa način funkcionisanja GPSS/FON programa se razlikovao od sadašnjeg. Tvorac GPSS/FON-a prof. dr Božidar Radenković je u vreme kada je implementirao prvu verziju programa imao pred sobom velika ograničenja resursa, naročito po pitanju slobodne memorije. Ranije je datoteka u kojoj se nalazio kod programa bila podeljena na veliki broj malih datoteka koje su se sve uključivale u jednu u toku izvršavanja direktivom $Include. U najnovijoj verziji su sve ove datoteke spojene u jednu.
U ranijim verzijama princip simulacije je bio sledeći:

1. Prvo bi se pokretala procedura za asembliranje koja je parsirala uneti kod na osnovu koga je kreirala obj verziju programa(u prvim verzijama programa obj program se snimao na disk, a u v 3.04 u memoriju).

2. Nakon nje pokretala se procedura za procesiranje koja je učitavala obj verziju programa u memoriju i vršila simulaciju.

U GPSS/FON v 4.0 je ovaj postupak je promenjen, tako da se sada nakon faze asembliranja ne kreira obj verzija programa. U najnovijoj verziji procedura za procesiranje koristi metodu prepareSimulation koja podatke iz faze asembliranja prenosi u fazu procesiranja. Ovim je izbegnuto zauzimanje nepotrebnih resursa i značajno pojednostavljen način rada programa. U ovom delu rada je otkrivena i jedna greška koja je postojala u radu ranijih verzija. Greška je bila u tome da priliko kreiranje obj verzije program je vršen ispis string – ova u obj verziju pri čemu se vršilo pretvaranje realnih brojeva u string. Prilikom ove konverzije na tri decimale dolazilo je do gubljenja određenih informacija tako da u pojedinim slučajevima, program nije radio potpuno tačno(sim1, sim1a, wash). U novoj verziji je ovaj nedostatak izbegnut jer se ne kreira obj verzija programa. U radu je prikazan source kod procedure prepareSimulation.
unit UPrepareSimulation;//*BojanNenadJovicic jul, avgust 2003.

                         {Unit u kome se nalazi implementacija procedure

                         prepareSimulation koja zamenjuje obj datoteku tj,

                         mObj(StringListu}

interface

        procedure prepareSimulation;

implementation

uses sysutils, UAsm, UGpss87;

procedure prepareSimulation;//*BojanNenadJovicic jul, avgust 2003.

        {procedura koja treba da zameni proceduru writeObj iz asemblera koja

        koja upisuje u stringListu(ranije u datoteku) i proceduru Input_ph iz

        procesora koja ucitava vrednosti iz string liste( datoteke) u memoriju

        da bih se pokrenuo procesor i izvrsio simulaciju

        }

var labb : longint;

PROCEDURE prepare_op (VAR Op : Operand);

BEGIN {prepare_op}

 inc(ocounter);

 case UAsm.STMTBL[counter]^.op[ocounter].typop of

   numop:  begin

             Op.Typ := Num_op;

             Op.N_op := UAsm.STMTBL[counter]^.op[ocounter].nop;

           end;

   snaop:  begin

             Op.Typ := Sna_op;

             Op.S_op.S_cod := abs(ord(UAsm.STMTBL[counter]^.op[ocounter].sop.snacod));

             case UAsm.STMTBL[counter]^.op[ocounter].sop.snaadr.adrs of

               UAsm.direct: begin

                                     Op.S_op.S_adr.Typ := Direct;

                                     Op.S_op.S_adr.Adrs := UAsm.STMTBL[counter]^.op[ocounter].sop.snaadr.nr;

                                   end;

               UAsm.indirect: begin

                                       Op.S_op.S_adr.Typ := Indirect;

                                       Op.S_op.S_adr.Adrs := UAsm.STMTBL[counter]^.op[ocounter].sop.snaadr.par;

                                     end;

             end;

           end;

   adrop:  begin

             Op.Typ := Adr_op;

             case UAsm.STMTBL[counter]^.op[ocounter].aop.adrs of

               UAsm.direct: begin

                                     Op.A_op.Typ := Direct;

                                     Op.A_op.Adrs := UAsm.STMTBL[counter]^.op[ocounter].aop.nr;

                                   end;

               UAsm.indirect: begin

                                       Op.A_op.Typ := Indirect;

                                       Op.A_op.Adrs := UAsm.STMTBL[counter]^.op[ocounter].aop.par;

                                     end;

             end;

           end;

 end;//case

end; {prepare_op}

function prepareOperand(o : UAsm.operand) : UGpss87.Operand;

begin

  case o.typop of

   numop:  begin

             prepareOperand.Typ := Num_op;

             prepareOperand.N_op := o.nop;

           end;

   snaop:  begin

             prepareOperand.Typ := Sna_op;

             prepareOperand.S_op.S_cod := abs(ord(o.sop.snacod));

             case o.sop.snaadr.adrs of

               UAsm.direct: begin

                                     prepareOperand.S_op.S_adr.Typ := Direct;

                                     prepareOperand.S_op.S_adr.Adrs := o.sop.snaadr.nr;

                                   end;

               UAsm.indirect: begin

                                       prepareOperand.S_op.S_adr.Typ := Indirect;

                                       prepareOperand.S_op.S_adr.Adrs := o.sop.snaadr.par;

                                     end;

             end;

           end;

   adrop:  begin

             prepareOperand.Typ := Adr_op;

             case o.aop.adrs of

               UAsm.direct: begin

                                     prepareOperand.A_op.Typ := Direct;

                                     prepareOperand.A_op.Adrs := o.aop.nr;

                                   end;

               UAsm.indirect: begin

                                       prepareOperand.A_op.Typ := Indirect;

                                       prepareOperand.A_op.Adrs := o.aop.par;

                                     end;

             end;

           end;

 end;//case

end;{prepeareOperand}

PROCEDURE prepare_blk;

var i : longint;

BEGIN {prepare_blk}

 IF labb > round(Max_blk * 0.9) THEN Extend_Blk_Tbl;

 IF labb < 1 THEN Error (3) ELSE

  BEGIN

    IF Blk_tbl [labb] = NIL THEN NEW (Blk_tbl[labb]);

    initBlkPtr(Blk_Tbl[labb]);

    Blk_tbl [labb]^.Op_cod := Opc;

    // Dodato ord oko Opc zbog izmene, BJ 6/7/2003

    for i:= 1 to Nr_OP[ord(Opc)] do

     prepare_op (Blk_tbl[labb]^.Op[i]);

    Blk_tbl [labb]^.N := 0;

    Blk_tbl [labb]^.W := 0

  END

END; {prepare_blk}

PROCEDURE prepare_dcl;

procedure prepare_fun;

var  i : LONGINT;

begin  {prepare_fun}

  IF labb > round(Max_fun * 0.9) THEN Extend_Fun_Tbl;

  IF Fun_tbl [labb] = NIL THEN NEW (Fun_tbl [labb]);

     initFunPtr(Fun_Tbl [labb]);

     prepare_op (Op);

     Fun_tbl [labb]^.A := Op.S_op;

     prepare_op (Op);

     j := Op.N_op;

     IF ABS (j) > Max_point THEN Error (15) ELSE

           BEGIN

             IF j > 0 THEN Fun_tbl [labb]^.Typ := Con ELSE

                              Fun_tbl [labb]^.Typ := Dis;

             Fun_tbl [labb]^.N := ABS (j);

             FOR i := 1 TO ABS (j) DO

              BEGIN

                Fun_tbl[labb]^.P[i].X := FUNTBL[STMTBL[counter]^.nr].point[i].x;//mObj.READ (Fun_tbl [labb]^.P[i].X);

                // Read(Ob_f, Fun_tbl [labb]^.P[i].X);

                Fun_tbl[labb]^.P[i].Y := FUNTBL[STMTBL[counter]^.nr].point[i].y;//mObj.READ (Fun_tbl [labb]^.P[i].Y);

                // Read(Ob_f, Fun_tbl [labb]^.P[i].Y);

              END

           END

END;{prepare_fun}

PROCEDURE prepare_ini;

VAR R : REAL; i : LONGINT;

BEGIN {prepare_ini}

 prepare_op (Op);

 Op_val (Op,R);

 IF R > round (Max_sav * 0.9)

 THEN Extend_Sav_Tbl;

     i := TRUNC(R);

     prepare_op (Op);

     Op_val (Op,R);

     Sav_tbl [i] := TRUNC(R)

END; {prepare_ini}

PROCEDURE prepare_tab;

VAR i: LONGINT;

BEGIN {prepare_tab}

  IF labb > round(Max_tab * 0.9) THEN Extend_Tab_Tbl;

    IF Tab_tbl [labb] = NIL THEN NEW (Tab_tbl [labb]);

    initTabPtr(Tab_Tbl [labb]);

    Tab_tbl [labb]^.Tb := 0;

    Tab_tbl [labb]^.Tc := 0;

    Tab_tbl [labb]^.Td := 0;

    Tab_tbl [labb]^.Sa := 0;

    Tab_tbl [labb]^.Avo := 0;

    prepare_op (Op);

    Tab_tbl [labb]^.Arg := OP.S_op;

    prepare_op (Op);

    Tab_tbl [labb]^.I [1].UL := OP.N_op;

    prepare_op (Op);

    Tab_tbl [labb]^.Si := Op.N_op;

    prepare_op (Op);

    Tab_tbl [labb]^.Ni := Op.N_op;

    j := Op.N_op + 1;

    IF Op.N_op > Max_intr - 1 THEN Error (12) ELSE

          BEGIN

            FOR i:= 1 TO Tab_tbl [labb]^.Ni + 1 DO

             BEGIN

              IF i > 1 THEN

               Tab_tbl [labb]^.I [i].UL := Tab_tbl [labb]^.I[1].Ul

                           + Tab_tbl [labb]^.Si * (i-1);

              Tab_tbl [labb]^.I[i].Off := 0

             END

          END

END; {prepare_tab}

PROCEDURE prepare_var;

VAR i : LONGINT ;

BEGIN {prepare_var}

 IF labb > round(Max_var  * 0.9) THEN Extend_Var_Tbl;

   IF Var_tbl [labb] = NIL THEN NEW (Var_tbl [labb]);

   initVarPtr(Var_Tbl [labb]);

   prepare_op (Op);

   Var_tbl [labb]^.N := Op.N_op;

   IF Op.N_op > Max_vop THEN Error (8) ELSE

     FOR i := 1 TO Op.N_op DO

       BEGIN

         if ( VARTBL[STMTBL[counter]^.nr].op[i]^.typ = UAsm.voperator ) then

         begin

           Var_tbl [labb]^.P_N [i].Typ := Voperator;

           Var_tbl [labb]^.P_N [i].Vopr := VARTBL[STMTBL[counter]^.nr].op[i]^.oprtr;

         end

           else

           begin

             Var_tbl [labb]^.P_N [i].Typ := Voperand;

             Var_tbl [labb]^.P_N [i].Vop := prepareOperand(VARTBL[STMTBL[counter]^.nr].op[i]^.oprnd);

           end;

       end;

 END; {prepare_var}

PROCEDURE prepare_sto;

VAR R : REAL;

BEGIN {prepare_sto}

  IF labb > round (Max_sto * 0.9)

        THEN Extend_Sto_Tbl;

   //BEGIN

     IF Sto_tbl [labb] = NIL THEN NEW (Sto_tbl [labb]);

     prepare_op (Op);

     Op_val (Op,R);

     Sto_tbl [labb]^.R := TRUNC(R);

     Sto_tbl [labb]^.S := 0;

     Sto_tbl [labb]^.Sa := 0;

     Sto_tbl [labb]^.Sc := 0;

     Sto_tbl [labb]^.Sr := 0;

     Sto_tbl [labb]^.Sm := 0;

     Sto_tbl [labb]^.St := 0;

     Sto_tbl [labb]^.Tht := 0;

     Sto_tbl [labb]^.C0 := 0

   //END

END; {prepare_sto}

BEGIN {prepare_dcl}

 CASE Opc OF         // Promenjeno iz brojeva u odgovarajuce enumerativne, BJ 6/7/2003

    ocRead_Fun : prepare_fun;

    ocRead_ini : prepare_ini;

    ocRead_sto : prepare_sto;

    ocRead_tab : prepare_tab;

    ocRead_var : prepare_var;

 END {CASE}

END; {prepare_dcl}

PROCEDURE prepare_ctr;

PROCEDURE Reset;

PROCEDURE Cec_rem (VAR Tr:Tra_ptr);

BEGIN  {Cec_rem}

  IF Tr^.Blink = NIL THEN

   BEGIN {Azurira se Head CEC}

    Cec.Head := Tr^.Flink;

    IF Cec.Head <> NIL THEN Cec.Head^.Blink  := NIL ELSE Cec.Tail := NIL;

   END

    ELSE IF Tr^.Flink = NIL THEN

      BEGIN {Azurira se tail CEC}

        Cec.Tail := Tr^.Blink;

        IF Cec.Tail <> NIL THEN Cec.Tail^.Flink := NIL ELSE Cec.Head := NIL;

      END

       ELSE

        BEGIN {Azuriraju se pokazivaci prethodne i sledece trans}

          TR^.Flink^.Blink := Tr^.Blink;

          TR^.Blink^.Flink := Tr^.Flink

        END

END; {Cec_rem}

PROCEDURE Fec_ins ( VAR  Tr : Tra_ptr );

{Ubacuje transakciju na koju pokazuje pokazivac Tr

 u FEC uredjenu po Bdt,Pr }

VAR Next,Prev : Tra_ptr; { Pointeri na predhodnu i sledecu

                           transakciju u FEC }

    Flg : BOOLEAN;

BEGIN {Fec_ins}

  IF Fec.Head = NIL THEN

   BEGIN {Ubacivanje transakcije u praznu listu}

     Fec.Head := Tr;

     Fec.Tail := Tr;

     Tr^.Flink := NIL;

     Tr^.Blink := NIL

   END  ELSE

   BEGIN

     Prev := Fec.Tail;

     Next := NIL;

    {Odredjivanje mesta za Tr u FEC}

     Flg := TRUE;

     WHILE  (Prev <> NIL) AND Flg DO

      BEGIN

       Flg := ((Prev^.Bdt > Tr^.Bdt) OR

            ((Prev^.Bdt = Tr^.Bdt) AND (Prev^.Pr < Tr^.Pr)));

       IF Flg THEN

        BEGIN

          Next := Prev;  {Prelazak na sledecu transakciju}

          Prev := Prev^.Blink

        END;

      END;

     IF Prev = NIL THEN

      BEGIN  {Ubacivanje na pocetak FEC}

        Fec.Head := Tr;

        Next^.Blink := Tr;

        Tr^.Flink := Next;

        Tr^.Blink := NIL;

      END ELSE

      IF Next = NIL THEN

       BEGIN {Ubacivanje na kraj FEC}

         Fec.Tail := Tr;

         Prev^.Flink := Tr;

         Tr^.Flink := NIL;

         Tr^.Blink := Prev

       END

       ELSE

        BEGIN

          Tr^.Flink := Next;

          Tr^.Blink := Prev;

          Prev^.Flink := Tr;

          Next^.Blink := Tr

        END

   END;

   Tr^.Ch := FEC_ch

END; {Fec_ins}

PROCEDURE Fec_rem (VAR Tr:Tra_ptr);

BEGIN  {Fec_rem}

  IF Tr^.Blink = NIL THEN

   BEGIN {Azurira se Head FEC}

    Fec.Head := Tr^.Flink;

    IF Fec.Head <> NIL THEN  Fec.Head^.Blink  := NIL ELSE Fec.Tail := NIL;

   END

   ELSE IF Tr^.Flink = NIL THEN

    BEGIN {Azurira se tail FEC}

      Fec.Tail := Tr^.Blink;

      IF Fec.Tail <> NIL THEN Fec.Tail^.Flink := NIL ELSE Fec.Head := NIL;

    END ELSE

     BEGIN {Azuriraju se pokazivaci prethodne i sledece trans}

      Tr^.Flink^.Blink := Tr^.Blink;

      Tr^.Blink^.Flink := Tr^.Flink

     END

END; {Fec_rem}

VAR i,j : LONGINT;

    Trans_tmp : Tra_ptr;

BEGIN {Reset}

 Trans_tmp := FEC.Head;

 WHILE Trans_tmp <> NIL DO  {Podesavanje vremena u FEC}

  BEGIN

    Trans_tmp^.Bdt := Trans_tmp^.Bdt - C1;

    Trans_tmp := Trans_tmp^.Flink

  END;

 WHILE Cec.head <> NIL DO {Anuliranje vremena u CEC i prebacivanje u FEC}

  BEGIN

    Trans_tmp := CEC.Head;

    Trans_tmp^.Bdt := 1;

    Cec_rem(Trans_tmp);

    Fec_ins(Trans_tmp)

  END;

 Trans_tmp := FEC.Head;

 WHILE Trans_tmp <> NIL DO

  BEGIN

    Trans := Trans_tmp;

    Trans_tmp := Trans_tmp^.Flink;

    // Enumerativni umesto 9, BJ 6/7/2003

    IF Blk_tbl[Trans^.Cb]^.Op_cod = ocError21 THEN

      BEGIN

       Fec_rem(Trans); {Generate}

       Dispose(Trans);

       Trans := NIL

      END;

  END;

 Trans := NIL;

 C1 := 0; {Anuliranje sata}

 {Brisanje statistike}

 FOR i:=1 TO Max_blk DO IF Blk_tbl[i] <> NIL THEN Blk_tbl[i]^.N:=Blk_tbl[i]^.W;

 FOR i := 1 TO Max_cha DO

  IF Cha_tbl [i] <> NIL THEN

   BEGIN

     Cha_tbl [i]^.Ca := 0;

     Cha_tbl [i]^.Cc := Cha_tbl [i]^.Ch;

     Cha_tbl [i]^.Cm := 0;

     Cha_tbl [i]^.Ct := 0;

     Cha_tbl [i]^.Tht := 0;

     Cha_tbl [i]^.C0 := 0;

   END;

 FOR i := 1 TO Max_fac DO

  IF Fac_tbl [i] <> NIL THEN

   BEGIN

     Fac_tbl [i]^.Fc := Fac_tbl [i]^.F;

     Fac_tbl [i]^.Fr := 0;

     Fac_tbl [i]^.Ft := 0;

     Fac_tbl [i]^.C0 := 0;

     Fac_tbl [i]^.Tht := 0;

   END;

 FOR i := 1 TO Max_sto DO

  IF Sto_tbl [i] <> NIL THEN

   BEGIN

     Sto_tbl [i]^.Sa := 0;

     Sto_tbl [i]^.Sc :=  Sto_tbl [i]^.S;

     Sto_tbl [i]^.Sr := 0;

     Sto_tbl [i]^.Sm := 0;

     Sto_tbl [i]^.St := 0;

     Sto_tbl [i]^.Tht := 0;

     Sto_tbl [i]^.C0 := 0

   END;

 FOR i := 1 TO Max_tab DO

  IF Tab_tbl [i] <> NIL THEN

   BEGIN

     Tab_tbl [i]^.Tb := 0;

     Tab_tbl [i]^.Tc := 0;

     Tab_tbl [i]^.Td := 0;

     Tab_tbl [i]^.Sa := 0;

     Tab_tbl [i]^.Avo:= 0;

     FOR j := 1 TO Max_intr DO Tab_tbl [i]^.I [j].Off := 0;

  END;

 FOR i := 1 TO Max_que DO

  IF Que_tbl [i] <> NIL THEN

   BEGIN

     Que_tbl [i]^.Qa := 0;

     Que_tbl [i]^.Qc := Que_tbl [i]^.Q;

     Que_tbl [i]^.Qm := 0;

     Que_tbl [i]^.Qt := 0;

     Que_tbl [i]^.Qx := 0;

     Que_tbl [i]^.Qz := 0;

     Que_tbl [i]^.Tht := 0;

     Que_tbl [i]^.Thz := 0;

     Que_tbl [i]^.C0 := 0;

     Que_tbl [i]^.Cz := 0

   END;

END; {Reset}

PROCEDURE Clear;

VAR S : ARRAY [1..5] OF word;

    i : LONGINT;

PROCEDURE Free (VAR Head : Tra_ptr); {Oslobadja transakcije}

VAR Temp : Tra_ptr;

BEGIN {Free}

   WHILE Head <> NIL DO

     BEGIN

       Temp := Head^.Flink;

       IF Head^.P <> NIL THEN DISPOSE (Head^.P);

       DISPOSE (Head);

       Head := Temp

     END

END; {Free}

BEGIN {Clear}

  FOR i:= 1 TO 5 DO  {Ulaz opreanada za CLEAR}

   BEGIN

     prepare_op (Op);

     S [i] := Op.A_op.Adrs

   END;

  Reset; {Anuliranje sata i ciscenje statistike}

  {Anuliranje SNA koje RESET ne anulira}

  Free (CEC.Head); CEC.Tail := NIL; {Oslobadjanje CEC}

  Free (FEC.Head); FEC.Tail := NIL; {Oslobadjanje FEC}

  FOR i := 1 TO Max_cha DO          {Oslobadjanje UC }

   IF Cha_tbl[i] <> NIL THEN

    BEGIN

     Free (Cha_tbl[i]^.Head);

     Cha_tbl [i]^.Tail := NIL

    END;

{Oslobadjanje prijemptovanih transakcija }

  FOR i := 1 TO Max_fac DO IF Fac_tbl[i] <> NIL THEN Free (Fac_tbl[i]^.Head);

  FOR i := 1 TO Max_blk DO IF Blk_tbl [i] <> NIL

     THEN

       BEGIN

         Blk_tbl [i]^.W := 0;

         Blk_tbl [i]^.N := 0

       END;

  FOR i := 1 TO Max_cha DO IF Cha_tbl [i] <> NIL THEN Cha_tbl [i]^.Ch := 0;

  FOR i := 1 TO Max_fac DO IF Fac_tbl [i] <> NIL

     THEN BEGIN

            Fac_tbl [i]^.Sc := 0;

            Fac_tbl [i]^.F := 0;

            Fac_tbl [i]^.Pc := 0

          END;

  FOR i := 1 TO Max_sto DO IF Sto_tbl [i] <> NIL

     THEN Sto_tbl [i]^.S := 0;

  FOR i := 1 TO Max_que DO IF Que_tbl [i] <> NIL

     THEN Que_tbl [i]^.Q := 0;

  FOR i := 1 TO Max_sav DO {Anuliranje Savevalue}

   IF NOT (i IN [ S[1],S[2],S[3],S[4],S[5] ]) THEN Sav_tbl [i] := 0;

END; {Clear}

PROCEDURE Start;

BEGIN {Start}

    prepare_op (Op);

    Tc := Op.N_op;

    prepare_op (Op);

    IF Op.N_op = 0 THEN P_flg := FALSE ELSE P_flg := TRUE

END; {Start}

BEGIN {prepare_ctr}

  CASE opc OF // Promenjeno u odgovarajuce enumarativne, BJ 6/7/2003

   ocClear : Clear;

   ocStop : Stop := TRUE; {END Naredba}

   ocReset : Reset;

   ocSim_flg : Sim_flg := TRUE; {Simulate1}

   ocStart : Start;

  ELSE BEGIN END;

  END {CASE}

END; {prepare_ctr}

{}

var read_int : integer; // Znak za pomoc pri konverziji pri ucitavanju, BJ 6/7/2003

BEGIN {prepareSimulation}

  REPEAT    // iz 0 u ocNull, BJ 6/7/2003

    inc(counter);

    ocounter := 0;

    labb := 0; Opc := ocNull;

    labb := UAsm.STMTBL[counter]^.nr;//mObj.READ (labb);

    // Read(Ob_f, labb);

    read_int := UAsm.STMTBL[counter]^.oc;//Obj.READ (read_int);

    // Read(Ob_F, read_int);

    Opc := Int2OC(read_int);

    IF Opc IN [ ocAdvance,

                ocAssign,

                ocDepart,

                ocEnter,

                ocGate,

                ocError21,

                ocLeave..ocRelease,

                ocSavevalue,

                ocSeize,

                ocTabulate..ocUnlink] THEN prepare_blk

    ELSE

    IF Opc IN [ ocRead_fun,

                ocRead_ini,

                ocRead_sto,

                ocRead_tab,

                ocRead_var] THEN prepare_dcl ELSE

    IF Opc IN [ ocClear,

                ocStop,

                ocReset,

                ocSim_flg,

                ocStart] THEN prepare_ctr ELSE Error (1);

    //mObj.READLN;

    inc(krug);

     //snimi_blokove(inttostr(krug));

    // Readln(Ob_f);

  until STOP or ( ( not Stop ) and (Tc > 0))  or ( counter = UAsm.stmcount ); // EOF(ob_f);

  IF ( counter = UAsm.stmcount{mObj.EOF})//EOF(ob_F)

      and (Opc <> ocStop) then Error (14);

  // snimi_blokove(FloatToStr(Now));

END; {prepareSimulation}

end.

3. Korisnički interfejs
U najnovijoj verziji je implementiran user interface u kome korisnik može program da unese ubacivanje određenih simbola na osnovu koga se kreira blok dijagram, koji se kasnije prevodi u listing. Postoji takođe opcija da snimanja i učitavanja blok dijagrama.

Slika 1. Izgled ekrana na kome se unosi blok dijagram

Slika 2. Ekran na kome je unet blok dijagram

Rad u grafičkom okruženju je jednostavan i prijatan. Prelaskom preko bilo kod dugmenceta na opcijama korisnik dobija hint u kome je objašnjeno koja se akcija aktivira njegovim pritiskom.

Grafički interfejs je implementiran u dva unita-a. Prvi unit je Ugrafički simboli u kome su definisane grafičke klase simbola koje se koriste za iscrtavanje i prevođenje blok dijagrama. U njihovoj implementaciji korišćeni su koncepti OOP-a. Zahvaljujući virtuelnim metodama i konceptu polimorfizma svi objekti neke od klasa Tsimbol se čuvaju u listi pokazivača TList. Pošto su sve glavne metode virtuelne nema potrebe za kastovanjem(cast), već su svi pozivi za baznu klasu Tsimbol. U radu je prikazan unit UgrafickiSimboli i glavne procedure unit-a UformDijagram.

unit UGrafickiSimboli;//*BojanNenadJovicic jul, avgust 2003.

                      { Unit u kome su definisane klase tipa simbol

                      koje se iscrtavaju na ekranu, funkcionalnost je postignuta

                      definisanje virtualnih metoda}

interface

uses ExtCtrls, Classes, Graphics, StdCtrls, Controls;

Type

        TSimbol = class(TPanel)

                constructor Create(AOwner : TComponent);override;

                constructor CreateFromString(state : string; AOwner : TComponent);

                private

                  prefiks : string;

                  image : TImage;

                  pictureCenter : TPicture;

                  pictureLeft : TPicture;

                public

                  function getSimbolText:string;virtual;

                  function getType:string;virtual;abstract;

                  procedure setPrefiks(prefiks : string);

                  function getPrefiks:string;

                  procedure setPictureLeft;virtual;

                  procedure setPictureCenter;virtual;

                  function save: string;virtual;abstract;

                end;

        TSimbolGenerate = class(TSimbol)

                constructor Create(AOwner : TComponent);override;

                constructor CreateFromString(state : string; AOwner : TComponent);

                private

                  edText : TEdit;

                public

                  function getSimbolText:string;override;

                  function getType:string;override;

                  function save:string;override;

                end;

        TSimbolAdvance = class(TSimbol)

                constructor Create(AOwner : TComponent);override;

                constructor CreateFromString(state : string; AOwner : TComponent);

                private

                  edText : TEdit;

                public

                  function getSimbolText:string;override;

                  function getType:string;override;

                  function save:string;override;

                end;

        TSimbolTransfer = class(TSimbol)

                constructor Create(AOwner : TComponent);override;

                constructor CreateFromString(state : string; AOwner : TComponent);

                private

                  edText : TEdit;

                  edTransfer : TEdit;

                public

                  function getSimbolText:string;override;

                  function getType:string;override;

                  function getTransfer:string;

                  function save:string;override;

                end;

        TSimbolTerminate = class(TSimbol)

                constructor Create(AOwner : TComponent);override;

                constructor CreateFromString(state : string; AOwner : TComponent);

                private

                  edText : TEdit;

                public

                  function getSimbolText:string;override;

                  function getType:string;override;

                  procedure setPictureLeft;override;

                  procedure setPictureCenter;override;

                  function save:string;override;

                end;

        TSimbolStart = class(TSimbol)

                constructor Create(AOwner : TComponent);override;

                constructor CreateFromString(state : string; AOwner : TComponent);

                private

                  edText : TEdit;

                public

                  function getSimbolText:string;override;

                  function getType:string;override;

                  function save:string;override;

                end;

implementation

function StrExp (in_str : string; n : integer) : string;

var tmp_str : string;

    i : integer;

begin

    tmp_str := in_str;

    for i := length(tmp_str) to n do

        tmp_str := tmp_str + ' ';

    result := tmp_str;

end;

function getState( var state : string):string;

begin

  result := copy(state, 1, pos('~', state) - 1);

  delete(state, 1, pos('~', state));

end;

constructor TSimbol.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  pictureCenter := TPicture.Create;

  pictureLeft := TPicture.Create;

  image := TImage.Create(Self);

  image.Parent := Self;

  image.Align := alClient;

  prefiks := '';

  BevelOuter := bvNone;

  Color := clWhite;

end;

constructor TSimbol.CreateFromString(state : string; AOwner : TComponent);

var simbolType : string;

begin

  simbolType := getState(state);

  if ( simbolType = 'generate' ) then

    Self := TSimbolGenerate.CreateFromString(state, AOwner)

  else

    if (simbolType = 'advance') then

      Self := TSimbolAdvance.CreateFromString(state, AOwner)

    else

      if ( simbolType = 'transfer' ) then

        Self := TSimbolTransfer.CreateFromString(state, AOwner)

      else

        if ( simbolType = 'terminate' ) then

          Self := TSimbolTerminate.CreateFromString(state, AOwner)

        else

          if ( simbolType = 'start' ) then

            Self := TSimbolStart.CreateFromString(state, AOwner)

end;

procedure TSimbol.setPrefiks(prefiks : string);

begin

  Self.prefiks := prefiks;

end;

function TSimbol.getPrefiks:string;

begin

  result := prefiks;

end;

function TSimbol.getSimbolText:string;

begin

  result := StrExp(Self.getPrefiks, 6);

end;

procedure TSimbol.setPictureLeft;

begin

  image.Picture := pictureLeft;

end;

procedure TSimbol.setPictureCenter;

begin

  image.Picture := pictureCenter;

end;

{TSimbolGenerate}

constructor TSimbolGenerate.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  Self.Width := 141;

  Self.Height := 91;

  pictureCenter.LoadFromFile('./Slicice/Generate.bmp');

  pictureLeft.LoadFromFile('./Slicice/Generate.bmp');

  image.Picture := pictureCenter;

  edText := TEdit.Create(Self);

  edText.Parent := Self;

  edText.Top := 35;

  edText.Left := 27;

  edText.Width := 111;

end;

constructor TSimbolGenerate.CreateFromString(state : string; AOwner : TComponent);

var atribute : string;

begin

  Self.Create(AOwner);

  atribute := getState(state);

  setPrefiks(atribute);

  atribute := getState(state);

  edText.Text := atribute;

end;

function TSimbolGenerate.getSimbolText():string;

begin

  result := inherited getSimbolText;

  result := result + 'GENERATE ' + edText.Text;

end;

function TSimbolGenerate.getType:string;

begin

  result := 'generate';

end;

function TSimbolGenerate.Save:string;

begin

  result := 'generate~' + getPrefiks + '~' + edText.Text + '~';

end;

{TSimbolGenerate}

{TSimbolAdvance}

constructor TSimbolAdvance.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  Self.Width := 141;

  Self.Height := 91;

  pictureCenter.LoadFromFile('./Slicice/Advance.bmp');

  pictureLeft.LoadFromFile('./Slicice/Advance.bmp');

  image.Picture := pictureCenter;

  edText := TEdit.Create(Self);

  edText.Parent := Self;

  edText.Top := 30;

  edText.Left := 27;

  edText.Width := 90;

end;

constructor TSimbolAdvance.CreateFromString(state : string; AOwner : TComponent);

var atribute : string;

begin

  Self.Create(AOwner);

  atribute := getState(state);

  setPrefiks(atribute);

  atribute := getState(state);

  edText.Text := atribute;

end;

function TSimbolAdvance.getSimbolText():string;

begin

  result := inherited getSimbolText;

  result := result + 'ADVANCE ' + edText.Text;

end;

function TSimbolAdvance.getType:string;

begin

  result := 'advance';

end;

function TSimbolAdvance.save: string;

begin

    result := 'advance~' + getPrefiks + '~' + edText.Text + '~';

end;

{TSimbolAdvance}

{TSimbolTransfer}

constructor TSimbolTransfer.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  Self.Width := 141;

  Self.Height := 91;

  pictureCenter.LoadFromFile('./Slicice/Transfer.bmp');

  pictureLeft.LoadFromFile('./Slicice/Transfer.bmp');

  image.Picture := pictureCenter;

  edText := TEdit.Create(Self);

  edText.Parent := Self;

  edText.Top := 35;

  edText.Left := 55;

  edText.Width := 35;

  edTransfer := TEdit.Create(Self);

  edTransfer.Parent := Self;

  edTransfer.Top := 8;

  edTransfer.Left := 118;

  edTransfer.Width := 17;

end;

constructor TSimbolTransfer.CreateFromString(state : string; AOwner : TComponent);

var atribute : string;

begin

  Self.Create(AOwner);

  atribute := getState(state);

  setPrefiks(atribute);

  atribute := getState(state);

  edText.Text := atribute;

  atribute := getState(state);

  edTransfer.Text := atribute;

end;

function TSimbolTransfer.getSimbolText():string;

begin

  result := inherited getSimbolText;

  result := result + 'TRANSFER ' + edText.Text + ',' + edTransfer.Text;

end;

function TSimbolTransfer.getType:string;

begin

  result := 'transfer';

end;

function TSimbolTransfer.getTransfer:string;

begin

  result := edTransfer.Text;

end;

function TSimbolTransfer.save: string;

begin

    result := 'transfer~' + getPrefiks + '~' + edText.Text + '~' + getTransfer + '~';

end;

{TSimbolTransfer}

{TSimbolTerminate}

constructor TSimbolTerminate.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  Self.Width := 141;

  Self.Height := 91;

  pictureCenter.LoadFromFile('./Slicice/Terminate.bmp');

  pictureLeft.LoadFromFile('./Slicice/TerminateLeft.bmp');

  image.Picture := pictureCenter;

  edText := TEdit.Create(Self);

  edText.Parent := Self;

  edText.Top := 35;

  edText.Left := 107;

  edText.Width := 30;

end;

constructor TSimbolTerminate.CreateFromString(state : string; AOwner : TComponent);

var atribute : string;

begin

  Self.Create(AOwner);

  atribute := getState(state);

  setPrefiks(atribute);

  atribute := getState(state);

  edText.Text := atribute;

end;

function TSimbolTerminate.getSimbolText():string;

begin

  result := inherited getSimbolText;

  result := result + 'TERMINATE ' + edText.Text;

end;

function TSimbolTerminate.getType:string;

begin

  result := 'terminate';

end;

procedure TSimbolTerminate.setPictureCenter;

begin

  inherited setPictureCenter;

  edText.Top := 35;

  edText.Left := 107;

end;

procedure TSimbolTerminate.setPictureLeft;

begin

  inherited setPictureLeft;

  edText.Top := 35;

  edText.Left := 72;

end;

function TSimbolTerminate.save: string;

begin

    result := 'terminate~' + getPrefiks + '~' + edText.Text + '~';

end;

{TSimbolTerminate}

{TSimbolStart}

constructor TSimbolStart.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  Self.Width := 141;

  Self.Height := 91;

  pictureCenter.LoadFromFile('./Slicice/Start.bmp');

  pictureLeft.LoadFromFile('./Slicice/Start.bmp');

  image.Picture := pictureCenter;

  edText := TEdit.Create(Self);

  edText.Parent := Self;

  edText.Top := 55;

  edText.Left := 30;

  edText.Width := 80;

end;

constructor TSimbolStart.CreateFromString(state : string; AOwner : TComponent);

var atribute : string;

begin

  Self.Create(AOwner);

  atribute := getState(state);

  setPrefiks(atribute);

  atribute := getState(state);

  edText.Text := atribute;

end;

function TSimbolStart.getSimbolText():string;

begin

  result := inherited getSimbolText;

  result := result + 'START ' + edText.Text;

end;

function TSimbolStart.getType:string;

begin

  result := 'start';

end;

function TSimbolStart.save: string;

begin

  result := 'start~' + getPrefiks + '~' + edText.Text + '~';

end;

{TSimbolStart}

end.
procedure TFormDijagram.nacrtajDijagram();

var i, countTransfer : integer;

    simbol, simbolPrev : TSimbol;

    prevTransfer : boolean;

    simbolTransfer : TSimbolTransfer;

begin

  obrisiDijagram;

  countTransfer := 0;

  for i := 0 to ( listaSimbola.Count - 1 ) do

  begin

    simbol := listaSimbola.Items[i];

    simbol.Parent := sbDijagram;

    prevTransfer := false;

    if ( i <> 0 ) then

    begin

      simbolPrev := listaSimbola.Items[i-1];

      if ( simbolPrev.getType = 'transfer' ) then

      begin

        prevTransfer := true;

        simbolTransfer := listaSimbola.Items[i-1];

        simbol.setPictureLeft;

        simbol.Left := (FormDijagram.Width - 140 ) div 2 + 140;

        simbol.Top := (i - countTransfer -1) * 90;

        simbol.setPrefiks(simbolTransfer.getTransfer);

        inc(countTransfer);

      end;

    end;

    if ( not prevTransfer ) then

    begin

      simbol.setPictureCenter;

      simbol.Left := (FormDijagram.Width - 140 ) div 2;

      simbol.Top := (i - countTransfer ) * 90;

    end;

  end;

end;//nacrtajDijagram

procedure TFormDijagram.obrisiDijagram();

var i : integer;

    simbol : TSimbol;

begin

  for i := 0 to ( listaSimbola.Count - 1 ) do

  begin

    simbol := listaSimbola.Items[i];

    simbol.Parent := parentSimbola;

  end;

  sbDijagram.Destroy;

  sbDijagram := TScrollBox.Create(FormDijagram);

  sbDijagram.Parent := FormDijagram;

  sbDijagram.Align := alClient;

  sbDijagram.Color := clWhite;

end;//obrisiDijagram

procedure TFormDijagram.prevediDijagram;

var i, j, brojTransfera: integer;

    simbol, simbolPom, simbolPrev : TSimbol;

    prevTransfer : boolean;

    nizTransfera : Array[1..100] of integer;

begin

  AForm.kod.Clear;

  AForm.kod.Lines.Add('SIMULATE');

  brojTransfera := 0;

  //ispisivanje u kod gpss programa

  for i := 0 to ( listaSimbola.Count - 1) do

  begin

    simbol := listaSimbola.Items[i];

    prevTransfer := false;

    if ( i <> 0 ) then

    begin

      simbolPrev := listaSimbola.Items[i-1];

      if ( simbolPrev.getType = 'transfer' ) then

      begin

        prevTransfer := true;

        inc(brojTransfera);

        nizTransfera[brojTransfera] := i;

      end;

    end;

    if ( not prevTransfer ) then

    begin

      if (simbol.getType = 'start' ) then

      begin

        //naredbe koje idu iz transfera

        for j := 1 to ( brojTransfera ) do

        begin

          simbolPom := listaSimbola[nizTransfera[j]];

          AForm.kod.Lines.Add(simbolPom.getSimbolText);

        end;

        brojTransfera := 0;

      end;

      AForm.kod.Lines.Add(simbol.getSimbolText);

    end;

 end;

 AForm.kod.Lines.Add('END');

end;//prevediDijagram

procedure TFormDijagram.snimiDijagram(fileName : string);

var fDijagram : textFile;

    fLinija : string;

    simbol : TSimbol;

    i : integer;

begin

  assignFile(fDijagram, fileName);

  rewrite(fDijagram);

  for i := 0 to ( listaSimbola.Count - 1 ) do

  begin

    simbol := listaSimbola[i];

    fLinija := simbol.Save;

    writeln(fDijagram, fLinija);

  end;

  closeFile(fDijagram);

end;//snimiDijagram

procedure TFormDijagram.ucitajDijagram(fileName : string);

var fDijagram : textFile;

    fLinija : string;

    simbol : TSimbol;

begin

  assignFile(fDijagram, fileName);

  reset(fDijagram);

  while ( not Eof(fDijagram) ) do

  begin

    Readln(fDijagram, fLinija);

    simbol := TSimbol.CreateFromString(fLinija, FormDijagram);

    listaSimbola.Add(simbol);

  end;

  closeFile(fDijagram);

end;//ucitajDijagram

